


PROGRAM FILE



Memotech MTX PANEL Fill Utility

by Terry Trotter

The Memotech computers have a 'front panel' display of registers for assembly language program debugging. There are several commands available on this front panel, but one that is missing is a memory FILL command. Memotech has provided a call in the ROM routines to a RAM location (FEXPAND) to allow expansion of the available commands. If a jump is placed at this location to the code to be executed, this will add a new command.

The FILL command checks for a

valid command, in this case F, and if found, continues with the FILL command. Otherwise, it returns to the panel commands.

The program starts by transferring the rest of the program code to high memory, resetting the stack limit, setting up the jump, and returning to Basic. The utility is then safe in memory.

When the command has been activated by F, it will prompt for a start address, an end address, and a FILL value.

10 CODE

```
4007 PANEXT: LD DE, $F6F0      ;point to where we will be
400A          LD ($FA92), DE   ;change stack limit
400E          LD ($FA9F), DE   ;change panel expand bytes
4012          LD HL, FILL      ;point to the start
4015          LD BC, $27       ;how many bytes to move
4018          LDIR             ;move them !
401A          LD A, $C3        ;set up the jump to fill
401C          LD ($FA9E), A
```

```
401F          RET
4020 FILL:    LD A, ($FD7D)     ;was the last character a "F" ?
4023          CP "F"
4025          RET NZ           ;if not return
4026          RST 28           ;print Fill and get bytes
4027          DB $AB
4028          DB "Fil", $EC     ;last byte has bit 7 set to 1
402C          PUSH BC          ;save the start address on stack
402D          RST 28           ;print To and get bytes
402E          DB $AB
402F          DB "T", $EF      ;last byte has bit 7 set to 1
4031          PUSH BC          ;save the end address on stack
4032          RST 28           ;print "With" and get byte
4033          DB $AB
4034          DB "Wit", $E8     ;last byte has bit 7 set to 1
4038          LD A, C           ;fetch the byte into A
4039          POP HL           ;get the end address
403A          POP DE           ;get the start address
403B          AND A            ;clear carry flag
403C          SBC HL, DE        ;calculate how many bytes
403E          LD B, H           ;set up length in BC
403F          LD C, L
4040          LD H, D           ;set up start address in HL
4041          LD L, E
4042          LD (DE), A        ;fill the first byte
4043          INC DE            ;point to the next with DE
4044          LDIR             ;fill the rest
4046          RET
```

Symbols:

```
FILL      4020      PANEXT  4007
```